Ventana Ground Fault Board Analysis

P. McGill
v1.1.0, 30Sep19

The DTEC Ground Fault and Water Alarm board has an input resistor network that produces an output voltage Vf at point B in the circuit proportional to ground-fault resistance Rf. See McGill's "general lab notebook" for equation derivation. Note that this is not the output of the circuit board—just the output of the input resistor network.




$$Vf = \frac{60}{2}\,\frac{\frac{Rf\,2}{Rf+2}}{2.2+\frac{Rf\,2}{Rf+2}}$$

$$\frac{60\,Rf}{(2+Rf)\left(2.2+\frac{2\,Rf}{2+Rf}\right)}$$

```
Plot[Vf, {Rf, 0.001, 10}, GridLines → Automatic,
 Frame -> True, FrameLabel → {"Fault Resistance (MΩ)", "Output (V)"}]
```



Above is the output of the resistor network before the voltage goes through the buffer and isolation amps. The AD202K isolation amp input range is +/- 5 V, so it will limit the max output voltage to 5 V.

```
Plot[Vf, {Rf, 0.001, 10}, PlotRange → {{0, 0.6}, {0, 5}}, GridLines → Automatic,
 Frame -> True, FrameLabel → {"Fault Resistance (MΩ)", "Output (V)"}]
```




The AD202K isolation amp is set up as an inverting amplifier, so the output voltage actually drops as the fault resistance increases. It is only by adjusting the offset potentiometer P1 to produce an offset of, say, +5 V that the board output can remain positive.

```
Plot[-Vf + 5, {Rf, 0.001, 10}, PlotRange → {{0, 0.6}, {0, 5}}, GridLines → Automatic,
 Frame -> True, FrameLabel → {"Fault Resistance (MΩ)", "Output (V)"},
 PlotLabel → "Output Voltage as a Function of Fault Resistance"]
```




According to the board setup procedure from DTEC, the board should produce 5.00 V with a 600 kΩ fault (offset adjustment P1), and 0 V with a 5 MΩ fault (gain adjustment P2). From these numbers we can find the required offset and gain.

To get a sense of what's going on, find the output of the input resistor network when the fault is 600 kΩ (0.6 MΩ), and 5 MΩ.

$$V0p6 = \frac{60}{2}\,\frac{\frac{0.6\times 2}{0.6+2}}{2.2+\frac{0.6\times 2}{0.6+2}}$$

5.20231

$$V5 = \frac{60}{2}\,\frac{\frac{5\times 2}{5+2}}{2.2+\frac{5\times 2}{5+2}}$$

11.811

Define a new function that defines the board output voltage in terms of the offset voltage Voff, the gain G, and the fault resistance Rf.

$$Vout[Voff\_, G\_, Rf\_] := Voff + G\,\frac{60}{2}\,\frac{\frac{Rf\ 2}{Rf+2}}{2.2+\frac{Rf\ 2}{Rf+2}}$$

We should be able to reproduce the third plot on this page if we set Voff = 5 and G = -1.

```
Plot[Vout[5, -1, Rf], {Rf, 0.001, 10},
 PlotRange → {{0, 0.6}, {0, 5}}, GridLines → Automatic, Frame -> True,
 FrameLabel → {"Fault Resistance (MΩ)", "Output (V)"},
 PlotLabel → "Output Voltage as a Function of Fault Resistance"]
```




Find the offset and gain that produces the calibration values.

```
soln = NSolve[{Vout[Voff, G, 0.6] == 5, Vout[Voff, G, 5] == 0.62}, {Voff, G}]
```

```
{{Voff → 8.44789, G → -0.662762}}
```

Now we can define a new output function that incorporates the calculated offset and gain values.

$$\text{Vout[Rf\_] := Voff} + \text{G}\,\frac{60}{2}\,\frac{\frac{\text{Rf}\,2}{\text{Rf}+2}}{2.2+\frac{\text{Rf}\,2}{\text{Rf}+2}}\ \text{/. First@soln}$$

```
plt = Plot[Vout[Rf], {Rf, 0.6, 5}, PlotRange → {{0, 5}, {0, 5}}, GridLines → Automatic,
  Frame -> True, FrameLabel → {"Fault Resistance (MΩ)", "Output (V)"},
  PlotLabel → "Output Voltage as a Function of Fault Resistance"]
```




Input the measured voltages produced by test resistances connected to a properly adjusted board (data provided by Duane Thompson).

```
drtDat = {{0.6, 5.000}, {2.73, 1.608},
   {3.03, 1.410}, {3.6, 1.117}, {4.42, 0.796}, {5, 0.620}};
```

```
Grid[drtDat, Frame → All]
```

| | |
|---|---|
| 0.6 | 5. |
| 2.73 | 1.608 |
| 3.03 | 1.41 |
| 3.6 | 1.117 |
| 4.42 | 0.796 |
| 5 | 0.62 |

Plot the measured data on top of the calculated output curve.

```
Show[{plt, Graphics[{Red, PointSize[Large], Point[drtDat]}]}]
```




Find the test fault resistor that will produce a 2.5 V output (50% fault).

```
NSolve[Vout[Rf] == 2.5, Rf] // First
```

```
{Rf → 1.77014}
```

This is 1.77 MΩ.

Find the inverse function that outputs resistance given the voltage and make a table that tells you what resistance produced the fault percentage shown on the *Ventana* GUI.

```
soln
```

```
{{Voff → 8.44789, G → -0.662762}}
```

$$\texttt{soln2 = NSolve}\left[\texttt{Vout == 8.448 + -0.6627} \times \frac{60}{2}\,\frac{\frac{\texttt{Rf 2}}{\texttt{Rf+2}}}{2.2 + \frac{\texttt{Rf 2}}{\texttt{Rf+2}}},\ \texttt{Rf}\right]\ \texttt{// First}$$

$$\left\{\texttt{Rf} \to -\frac{29.3333\ (-1056. + 125.\ \texttt{Vout})}{3567. + 3500.\ \texttt{Vout}}\right\}$$

```
Grid[Prepend[Table[{Vout, 20 Vout, NumberForm[soln2[[1, 2]], 3]}, {Vout, 0, 5, 0.5}],
  {"Vout", "% FS", "MΩ"}], Frame → All]
```

| Vout | % FS | MΩ |
|---|---|---|
| 0. | 0. | 8.68 |
| 0.5 | 10. | 5.48 |
| 1. | 20. | 3.86 |
| 1.5 | 30. | 2.89 |
| 2. | 40. | 2.24 |
| 2.5 | 50. | 1.77 |
| 3. | 60. | 1.42 |
| 3.5 | 70. | 1.15 |
| 4. | 80. | 0.928 |
| 4.5 | 90. | 0.749 |
| 5. | 100. | 0.6 |